Examples about looping in Ada
1.

**************************************

with Ada.Text_IO; use Ada.Text_IO;
with Ada.Integer_Text_IO; use Ada.Integer_Text_IO;
procedure For_Loop is

begin

  for ACounter in 3..8 loop
    Put(Item => ACounter, Width => 3);
  end loop;
  New_Line;

end For_Loop;

2.

***************************************

with Ada.Text_IO; 
procedure Input_Loop is
---------------------------------------------------------------
--| Robust interactive input loop - handle exceptions
--| Michael B. Feldman; last revised Sept. 1999
---------------------------------------------------------------

  type Days is (Mon,Tue,Wed,Thu,Fri,Sat,Sun);
  package Day_IO is 
    new Ada.Text_IO.Enumeration_IO(Enum => Days); 
  Day1: Days;

begin

  loop
    begin
      Ada.Text_IO.Put(Item => "Please enter a Day >"); 
      Day_IO.Get(Item => Day1);
      Ada.Text_IO.Put(Item => "Day = "); 
      Day_IO.Put(Item => Day1); 
      Ada.Text_IO.New_Line;
      exit;
    exception
      when Ada.Text_IO.Data_Error =>
        Ada.Text_IO.Put_Line(Item => "Bad data; try again"); 
        Ada.Text_IO.Skip_Line; -- flush input line
    end;
  end loop;

  Ada.Text_IO.Put_Line
    (Item => "Good input; here we'd continue processing");

exception

  when Ada.Text_IO.Data_Error => 
    Ada.Text_IO.Put_Line(Item => "We got a data error"); 
  when Ada.Text_IO.End_Error  => 
    Ada.Text_IO.Put_Line(Item => "Tried to read past EOF"); 
  when others => 
    Ada.Text_IO.Put_Line
      (Item => "Not sure why we got here"); 

end Input_Loop;
********************************************

3.

********************************************

with Ada.Text_IO; use Ada.Text_IO;
with Ada.Integer_Text_IO; use Ada.Integer_Text_IO;
procedure More_Loops is

  ACounter: Integer;

begin

  for ACounter in 3..8 loop
    Put(Item => ACounter, Width => 3);
  end loop;
  New_Line;

  ACounter := 3;
  while ACounter <= 8 loop 
    Put(Item => ACounter, Width => 3);
    ACounter := ACounter + 1;
  end loop;
  New_Line;

end More_Loops; 
*********************************

4.

*********************************

with Ada.Text_IO; use Ada.Text_IO;
with Ada.Integer_Text_IO; use Ada.Integer_Text_IO;
procedure Nested_Loop_with_Subtype is

  subtype Little is Integer range -5..0;

begin

  for OuterCounter in Little loop
    Put(Item => OuterCounter, Width => 3);
    Put(" |");

    for InnerCounter in reverse Little loop
      Put(Item => OuterCounter + InnerCounter, Width => 3);
    end loop;

    New_Line;
  end loop;

end Nested_Loop_with_Subtype;
*********************************************

5.

*********************************************

with Ada.Text_IO; use Ada.Text_IO;
with Ada.Integer_Text_IO; use Ada.Integer_Text_IO;
procedure Still_More_Loops is

  ACounter: Integer;
  CharCounter: Character;

begin

  ACounter := 3;
  loop
    Put(Item => ACounter, Width => 3);
    exit when ACounter = 8;
    ACounter := ACounter + 1;
  end loop;
  New_Line;

  CharCounter := 'f';
  loop
    Put("  " & CharCounter);
    exit when CharCounter = 'k';
    CharCounter := Character'Succ(CharCounter);
  end loop;
  New_Line;

end Still_More_Loops; 
***********************************************

6.

with Ada.Text_IO; use Ada.Text_IO;
procedure Show_Function is
  type Day is (Mon,Tue,Wed,Thu,Fri,Sat,Sun);
  package Day_Io is new Enumeration_Io(Day); use Day_Io;
  Day1: Day;
  Day2: Day;

  function Max(Left, Right: Day) return Day is
  begin
    if Left >= Right then
      return Left;
    else
      return Right;
    end if;
  end Max;

begin
  loop
    Put("Please enter Day1 "); Get(Day1);
    Put("Please enter Day2 "); Get(Day2);

    Put("Day1 = "); Put(Day1); New_Line;
    Put("Day2 = "); Put(Day2);
    New_Line;
    Put("The larger of the two days = ");
    Put(Item=>Max(Day1,Day2),Width=>10);
    New_Line;
  end loop;
end Show_Function;
**************************************************
7.

with Ada.Text_IO; use Ada.Text_IO;
with Ada.Integer_Text_IO; use Ada.Integer_Text_IO;
procedure Show_Procedure is
  
  Number1, Number2: Integer;
  Number3: Integer;

  procedure Max(Left, Right: in Integer; Result: out Integer) is
  begin
    if Left >= Right then
      Result := Left;
    else
      Result := Right;
    end if;
  end Max;

begin
  loop
    Put("Please enter integer Number1 "); Get(Number1);
    Put("Please enter integer Number2 "); Get(Number2);

    Put("Number 1 ="); Put(Number1); New_Line;
    Put("Number 2 ="); Put(Number2); New_Line;
    Put("The larger of the two numbers =");
    Max(Number1,Number2,Number3);
    Put(Number3); New_Line;
    New_Line;
  end loop;
end Show_Procedure;

